tested 190817 using *SpinDynamica* 3.0.1 under *Mathematica* 11.0

init

```
Needs["SpinDynamica`"]
```

```
SetSpinSystem[2]
```

SetSpinSystem: the spin system has been set to $\{\{1, \frac{1}{2}\}, \{2, \frac{1}{2}\}\}$

set up Hamiltonian and DD relaxation superoperator

- **lab frame Hamiltonian (for thermalization purposes)**

```
H0lab = ω0 opI["z"]
```

$\omega0\ (I_{1z} + I_{2z})$

- **rotating frame Hamiltonian (for spin-dynamical purposes)**

```
H0 = Ω1 opI[1, "z"] + Ω2 opI[2, "z"] + 2 π J12 opI[1].opI[2]
```

$2\ J12\ \pi\ (I_{1x} \cdot I_{2x} + I_{1y} \cdot I_{2y} + I_{1z} \cdot I_{2z}) + \Omega1\ I_{1z} + \Omega2\ I_{2z}$

- **relaxation superoperator for DD relaxation**

```
SpecJ[ω0_, τc_] := τc / (1 + ω0^2 τc^2)
```

```
ΓDD = - (6 / 5) b12^2 Sum[
    (-1)^m SpecJ[m ω0, τc] ×
     DoubleCommutationSuperoperator[opT[{1, 2}, {2, m}], opT[{1, 2}, {2, -m}]],
    {m, -2, 2}];
```

- **thermalized relaxation superoperor:**

```
ΓDDtherm = ThermalizeSuperoperator[ΓDD, H0lab, 300];
```

parameters

```
parameters =.
```

```
?PhysicalConstantValues
```

PhysicalConstantValues provides the substitution rules for the SI
numerical values of selected physical constants *c*, *h*, *ℏ*, *μ*0, *ϵ*0, kB, Nav, *e*, me, mp, mn

```
parameters[corrtime_] := {ω0 → 2 π (-600 × 10^6), Ω1 → 2 π (-2×10^3), Ω2 → 2 π (4×10^3), J12 → 15,
   ωnut → 2 π 20 , b12 → 2 π (-30 × 10^3), τc → corrtime, T → 300, Sequence @@ PhysicalConstantValues}
```

spectral simulation

- **20 ps correlation time**

```
ListPlot[
 Re@FT@Signal1D[{0, 1, 1/512}, N[Liouvillian[H0, ΓDDtherm] /. parameters[20×10^-12]]],
 PlotRange → All, Frame → True, Joined → True]
```

Signal1D: Using SignalCalculationMethod → Diagonalization

Signal1D: the last sampling point has been dropped in order to get an even number of points.

Signal1D: Using LineBroadening → 2π × 1.46587 rad s^{-1}.



- **1 ns correlation time**

```
ListPlot[Re@FT@Signal1D[{0, 1, 1/512}, N[Liouvillian[H0, ΓDDtherm] /. parameters[10^-9]]],
 PlotRange → All, Frame → True, Joined → True]
```

Signal1D: Using SignalCalculationMethod → Diagonalization

Signal1D: the last sampling point has been dropped in order to get an even number of points.

Signal1D: Using LineBroadening → 2π × 1.46587 rad s^{-1}.



■ **3 ns correlation time**

```
ListPlot[Re@FT@Signal1D[{0, 1, 1/512}, N[Liouvillian[H0, ΓDDtherm] /. parameters[3×10^-9]]],
 PlotRange → All, Frame → True, Joined → True]
```

Signal1D: Using SignalCalculationMethod → Diagonalization

Signal1D: the last sampling point has been dropped in order to get an even number of points.

Signal1D: Using LineBroadening → 2π × 1.46587 rad s^{-1}.



trajectories of z-magnetizations with selective inversion, displaying cross-relaxation effects

■ **10 ps correlation time**

```
corrtime = 10 × 10^-12;

ρeq = ThermalEquilibriumDensityOperator[
   H0lab /. parameters[corrtime], Temperature → T /. parameters[corrtime]]
```

```
Operator[<< .. >> , OperatorType → Hermitian ]
```

```
{trajz1, trajz2} =
 Trajectory[
   ρeq → {opI[1, "z"], opI[2, "z"]},
   {RotationSuperoperator[1, {π, "x"}], {None, 30}},
   BackgroundGenerator → N[(Liouvillian[H0, ΓDDtherm] /. parameters[corrtime])],
   NormalizationFactor → OperatorAmplitude[ρeq → opI[1, "z"]]
 ]
```

```
{TrajectoryFunction[ {{0, 30.}} , <>], TrajectoryFunction[ {{0, 30.}} , <>]}
```

```
Plot[
 Evaluate@Re[{trajz1[t], trajz2[t]}],
 {t, 0, 30},
 Frame → True,
 PlotRange → All,
 PlotStyle → {{Thick, Blue}, {Thick, Red}},
 LabelStyle → Directive[Medium, FontFamily → "Helvetica"]
]
```



note the transient enhancement of the magnetization of the non-irradiated spin. This is the transient NOE.

1 ns correlation time

```
corrtime = 1 × 10^-9;
```

```
ρeq = ThermalEquilibriumDensityOperator[
  H0lab /. parameters[corrtime], Temperature → T /. parameters[corrtime]]
```

```
Operator[<< .. >> , OperatorType → Hermitian ]
```

```
{trajz1, trajz2} = Trajectory[
  ρeq → {opI[1, "z"], opI[2, "z"]},
  {RotationSuperoperator[1, {π, "x"}], {None, 1.5}},
  BackgroundGenerator → N[(Liouvillian[H0, ΓDDtherm] /. parameters[corrtime])],
  NormalizationFactor → OperatorAmplitude[ρeq → opI[1, "z"]]
 ]
```

```
{TrajectoryFunction[ {{0, 1.5}} , <>], TrajectoryFunction[ {{0, 1.5}} , <>]}
```

```
Plot[
 Evaluate@Re[{trajz1[t], trajz2[t]}],
 {t, 0, 1.5},
 Frame → True,
 PlotRange → {-1.1, 1.5},
 PlotStyle → {{Thick, Blue}, {Thick, Red}},
 LabelStyle → Directive[Medium, FontFamily → "Helvetica"]
]
```



for long correlation time, the NOE is negative

simulate trajectories of z-magnetizations with rf field applied to one of the 2 spins (steady-state NOE)

UseDiagonalizationWhenPossible → True is used in this case, since the trajectories are rapidly oscillating and need to be calculated for a long time

- **10 ps correlation time**

```
corrtime = 10 × 10^-12;
```

```
parameters[corrtime]
```

$\{\omega 0 \to -1\,200\,000\,000\,\pi,\ \Omega 1 \to -4000\,\pi,\ \Omega 2 \to 8000\,\pi,\ J12 \to 15,\ \omega nut \to 40\,\pi,\ b12 \to -60\,000\,\pi,\ \tau c \to \frac{1}{100\,000\,000\,000},\ T \to 300,\ c \to 299\,792\,458,\ h \to 6.626070 \times 10^{-34},\ \hbar \to 1.0545717 \times 10^{-34},\ \mu 0 \to \frac{\pi}{2\,500\,000},\ \epsilon 0 \to \frac{625\,000}{22\,468\,879\,468\,420\,441\,\pi},\ kB \to 1.38065 \times 10^{-23},\ Nav \to 6.022141 \times 10^{23},\ e \to 1.6021766 \times 10^{-19},\ me \to 9.109383 \times 10^{-31},\ mp \to 1.672622 \times 10^{-27},\ mn \to 1.674927 \times 10^{-27}\}$

```
ωnut /. parameters[corrtime]
```

$40\,\pi$

```
{trajz1, trajz2} =
 Trajectory[
  ρeq → {opI[1, "z"], opI[2, "z"]},
  {ωnut opI[1, "x"] - Ω1 opI[1, "z"], 30} /. parameters[corrtime],
  BackgroundGenerator → N@(Liouvillian[H0, ΓDDtherm] /. parameters[corrtime]),
  NormalizationFactor → OperatorAmplitude[ρeq → opI[1, "z"]]
 ]
```

```
{TrajectoryFunction[ {{0, 30.}} , <>], TrajectoryFunction[ {{0, 30.}} , <>]}
```

```
Plot[
 Evaluate@Re[{trajz1[t], trajz2[t]}],
 {t, 0, 10},
 Frame → True,
 PlotRange → {-1.1, 1.5},
 PlotStyle → {{Thick, Blue}, {Thick, Red}},
 PlotPoints → 10,
 LabelStyle → Directive[Medium, FontFamily → "Helvetica"]
]
```



note that the magnetization of the second spin builds up as the magnetization of the first spin is saturated. Note also that the saturation is not complete, due to the limited rf field strength.

- **1 ns correlation time**

```
corrtime = 1 × 10^-9;
```

```
{trajz1, trajz2} = Trajectory[
  ρeq → {opI[1, "z"], opI[2, "z"]},
  {ωnut opI[1, "x"] - Ω1 opI[1, "z"], 10} /. parameters[corrtime],
  BackgroundGenerator → N@(Liouvillian[H0, ΓDDtherm] /. parameters[corrtime]),
  UseDiagonalizationWhenPossible → True,
  NormalizationFactor → OperatorAmplitude[ρeq → opI[1, "z"]]
 ]
```

```
{TrajectoryFunction[ {{0, 10.}} , <>], TrajectoryFunction[ {{0, 10.}} , <>]}
```

```
Plot[
 Evaluate@Re[{trajz1[t], trajz2[t]}],
 {t, 0, 10},
 Frame → True,
 PlotRange → All,
 PlotStyle → {{Thick, Blue}, {Thick, Red}}
]
```



note the partial saturation of the irradiated spin, while the magnetization of the other spin is reduced